

Chapter 2

Basic Multilevel Models

Birth Order Data Analysis.....	2-3
Random Effects ANOVA.....	2-3
Adding Lower-Level Predictors to the Model.....	2-9
Adding Upper-Level Predictors to the Model	2-11
Beta Interferon Analysis.....	2-13
Exploratory Data Analysis.....	2-13
Random-Effects ANOVA Model.....	2-16
Adding Upper-Level Predictors to the Model	2-18
Adding Lower-Level Predictors to the Model.....	2-20

Birth Order Data Analysis

For this demonstration we will again use the `birth.sas7bdat` data file (after deleting the outlier as shown in Chapter 1). Accompanying SAS code is in the `Birth_RE.sas` file.

Random Effects ANOVA

We can get a crude sense of how meaningful between-family differences in Math IQ are by computing and plotting the distribution of family mean IQ. We do this by first outputting the mean IQ for each family using PROC MEANS. We will use these means in two plots, an index plot and a histogram of family means. For the index plot, we merge the means back into the original data (producing the data set called `plot`).

```
proc sort data=birth; by mom_id; run;
proc means data=birth noprint; by mom_id;
  var math;
  output out=mean mean=mathm;
run;

proc print data=mean(obs=10); run;
data mean; set mean; index = _N_; run;
data plot; merge birth mean; by mom_id;
run;
proc print data=plot (obs=20);
  var mom_id index math _freq_ mathm;
run;
```

This is some of the data produced by PROC MEANS, now in the `means` data set:

Obs	mom_id	_TYPE_	_FREQ_	mathm
1	3	0	1	102
2	4	0	1	105
3	8	0	2	99.5
4	20	0	1	112
5	25	0	1	105
6	43	0	2	81
7	44	0	1	100
8	49	0	2	107.5
9	50	0	2	109.5
10	57	0	1	135

And these means are merged back into the original data in the `plot` data set:

Obs	mom_id	index	math	_FREQ_	mathm
1	3	1	102	1	102
2	4	2	105	1	105
3	8	3	102	2	99.5
4	8	3	97	2	99.5
5	20	4	112	1	112
6	25	5	105	1	105
7	43	6	97	2	81
8	43	6	65	2	81
9	44	7	100	1	100
10	49	8	126	2	107.5

11	49	8	89	2	107.5
12	50	9	111	2	109.5
13	50	9	108	2	109.5
14	57	10	135	1	135


Next, we produce our plots. Note that the index plot is made on a subset of the families in the data (the 56th through 75th families) and that we selected only families with 2+ children in the data for the histogram because otherwise the family mean and child score are identical.


```

goptions vsize=5 hsize=8 htext=1.25;
symbol1 value=dot h=.5 w=.5 color=black repeat=1;
symbol2 font=simplex value="--" color=black repeat=1;
axis1 minor=none label="Math";
axis2 minor=none offset=(2) label="Family" order=(56 to 75 by 1);
legend1 position=(inside bottom left) offset=(2) down=2 label=none
value=('Child' 'Mean');
proc gplot data=plot; where 55<index<=75;
  plot (math mathm)*index/legend=legend1 overlay vref=98.8 lvref=33
haxis=axis2 vaxis=axis1;
run;quit;

goptions hsize=5 vsize=4;
proc univariate data=mean; where _FREQ_>1;
  var mathm;
  histogram/ cfill=red kernel(color=black w=2) haxis=axis1;
  axis1 label="Family Mean Math Test Score";
run;

```





The index plot shows that there is both within- and between-family variation. There are individual differences among siblings within a family, but also differences in mean math IQ across families. Focusing on the across-family differences, the histogram shows a great deal of family-to-family variation in math IQ, and that this variation follows the form of a normal distribution. Of course, some of the variation is due to simple sampling error (since each family mean is computed from very few observations), so we would like to more formally evaluate within- versus between-family IQ differences. The random-effects ANOVA model provides a way to decompose this variance.

The RE-ANOVA model is

Level 1:

$$y_{ij} = \beta_{0j} + r_{ij} \quad r_{ij} \sim N(0, \sigma^2)$$

Level 2:

$$\beta_{0j} = \gamma_{00} + u_{0j} \quad u_{0j} \sim N(0, \tau_{00})$$

Reduced-Form:

$$y_{ij} = \gamma_{00} + u_{0j} + r_{ij}$$

The code for fitting this model is shown here:

```
proc mixed data=demo.birth method=reml covtest cl noclprint;
  class mom_id;
  model math = / solution ddfm=bw;
  random intercept / subject=mom_id;
run;
```

Syntax Notes:

- PROC MIXED statement: invokes the MIXED procedure
 - METHOD=REML tells MIXED to use the REstricted Maximum Likelihood Estimator to fit the model.
 - COVTEST produces z-tests for the variance parameters of the model.
 - CL produces confidence limits for the variance parameters of the model.
 - NOCLPRINT suppresses printing of all levels of variables declared in the CLASS statement.
- CLASS statement: Used to declare classification (nominal) variables used in the model specification.
 - We have listed the ID variable for the nesting (**mom_id**). It is not essential to include this here, but it can help to avoid errors (see description of RANDOM statement).
- MODEL statement: Used to define the outcome variable (left of =) and predictors with fixed effects in the model (right of =). MIXED will estimate a fixed effect for any predictor included in this statement. MIXED will estimate $G-1$ effects for any G -level categorical predictor defined previously in the CLASS statement. A fixed intercept is estimated by default.
 - SOLUTION tells SAS to report the table of fixed effect estimates, standard errors, and p-values.
 - DDFM tells SAS which method to use for computing degrees of freedom for testing fixed effects. BW is the heuristic “Between and Within” method which draws an analogy to repeated measures ANOVA for computing degrees of freedom.
- RANDOM statement: Used to define the random effects of the model. A random intercept is *not* included by default and must be defined by putting the keyword “Intercept.”
 - SUBJECT is used to indicate the nesting within the data (here, siblings within **mom_ID**). *If this ID variable is numeric and the data is sorted on this variable in advance of the MIXED procedure then it is not necessary that it be declared in the CLASS statement. But, if it is not sorted, results will be wrong.*

Results include some descriptive information:

Model Information	
Data Set	WORK.BIRTH
Dependent Variable	math
Covariance Structure	Variance Components
Subject Effect	mom_id
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within
Dimensions	
Covariance Parameters	2
Columns in X	1
Columns in Z Per Subject	1
Subjects	2207
Max Obs Per Subject	4
Number of Observations	
Number of Observations Read	3312
Number of Observations Used	3312
Number of Observations Not Used	0

Note that in the “Dimensions” table, SAS reports 2 (co)variance parameters, corresponding to σ^2 and τ_{00} . “Columns in X” refers to number of columns in the design matrix for the fixed effects (here just a single column of ones for the fixed intercept parameter), whereas “Columns in Z” refers to the number of columns in the design matrix for the random effects (also a single column of ones in this case, for the random intercept). “Subjects” reports the number of unique units declared in the SUBJECT option of the RANDOM statement (here, 2207 families).

The next bit of output tells us that the model converged. *Do not interpret your model estimates if the model fails to converge.*

Iteration History			
Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	26107.69744119	
1	3	25942.89945587	0.00000931
2	1	25942.80545103	0.00000001
Convergence criteria met.			

We are next provided with information about the variance parameters of the model:

Covariance Parameter Estimates							
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	Alpha	Lower
Intercept	mom_id	52.1217	4.3981	11.85	<.0001	0.05	44.4716
Residual		102.30	4.0334	25.36	<.0001	0.05	94.8366

Covariance Parameter Estimates		
Cov Parm	Subject	Upper
Intercept	mom_id	61.9420
Residual		110.69

The “Covariance Parameter Estimates” table provides estimates for the random intercept variance $\hat{\tau}_{00} = 52.12$ and the residual variance $\hat{\sigma}^2 = 102.30$. The corresponding ICC is $52.12 / (52.12 + 102.30) = .34$. The standard errors, z-values and p-values shown here for the variance estimates were obtained from the COVTEST option of the MIXED statement, whereas the lower and upper confidence limits for the 95% confidence intervals were obtained from the CL option of the MIXED procedure. *Note that the z-tests and confidence intervals are computed using different assumptions about the sampling distributions of the variance parameters; hence it is possible to find p-values for the z-test greater than .05 simultaneously with CIs that do not include zero.*

SAS also provides some fit statistics:

Fit Statistics	
-2 Res Log Likelihood	25942.8
AIC (smaller is better)	25946.8
AICC (smaller is better)	25946.8
BIC (smaller is better)	25958.2

We defer discussion of these until later.

Finally, we obtain the estimates for the fixed effects:

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	98.9115	0.2386	2206	414.50	<.0001

This table shows that the average value of Math IQ is 98.91. Note that this will not exactly equal the grand mean ignoring clustering because MIXED implicitly adjusts for the partial redundancy in data due to dependence. This table is produced only when the SOLUTION option is included on the MODEL statement.

Adding Lower-Level Predictors to the Model

We will now add fixed effects for birth order, cohort and birth order × cohort to the model at Level 1. Our model will now be:

Level 1:

$$Math_{ij} = \beta_{0j} + \beta_{1j}Second_{ij} + \beta_{2j}Third_{ij} + \beta_{3j}Fourth_{ij} + \beta_{4j}Old_{ij} + \beta_{5j}Second_{ij} \times Old_{ij} + \beta_{6j}Third_{ij} \times Old_{ij} + \beta_{7j}Fourth_{ij} \times Old_{ij} + r_{ij}$$

$$r_{ij} \sim N(0, \sigma^2)$$

Level 2:

$$\beta_{0j} = \gamma_{00} + u_{0j} \quad u_{0j} \sim N(0, \tau_{00})$$

$$\beta_{pj} = \gamma_{p0} \text{ for } p > 0$$

Reduced-Form:

$$Math_{ij} = \gamma_{00} + \gamma_{10}Second_{ij} + \gamma_{20}Third_{ij} + \gamma_{30}Fourth_{ij} + \gamma_{40}Old_{ij} + \gamma_{50}Second_{ij} \times Old_{ij} + \gamma_{60}Third_{ij} \times Old_{ij} + \gamma_{70}Fourth_{ij} \times Old_{ij} + u_{0j} + r_{ij}$$

To fit the model in PROC MIXED, we first sort the data in descending order by **brthordr** and **cohort** so that we can make the reference categories first born children and children in the younger cohort using the ORDER=DATA option.

```
proc sort data=birth; by descending brthordr descending cohort; run;
proc mixed data=birth method=reml covtest cl order=data noclprint;
  class brthordr cohort mom_id;
  model math = brthordr cohort cohort*brthordr / solution ddfm=bw;
  random intercept / subject=mom_id;
run;
```

Brthordr, **cohort**, and **mom_id** are both declared in the CLASS statement and then **brthordr**, **cohort**, and **cohort*brthordr** fixed effects are included by adding these predictors to the MODEL statement. Abridged results follow:

Dimensions	
Covariance Parameters	2
Columns in X	15
Columns in Z per Subject	1
Subjects	2207
Max Obs per Subject	4

In the “Dimensions” table, SAS reports 15 columns in X. These are 1 (intercept) + 4 (birth orders) + 2 (cohorts) + 4×2 (birth orders x cohorts). All levels of the CLASS variables are included in the count despite the fact that estimates are *not* produced for all of them (e.g., 3 dummy variables suffice to summarize the 4 levels of birth order).

Covariance Parameter Estimates								
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	Alpha	Lower	Upper
Intercept	mom_id	49.2755	4.2992	11.46	<.0001	0.05	41.8264	58.9196
Residual		101.61	4.0050	25.37	<.0001	0.05	94.1991	109.94

Solution for Fixed Effects								
Effect	brthordr	cohort	Estimate	Standard Error	DF	t Value	Pr > t	
Intercept			101.50	0.4742	2206	214.02	<.0001	
brthordr	4		-3.4931	1.2264	1098	-2.85	0.0045	
brthordr	3		-2.7351	0.8037	1098	-3.40	0.0007	
brthordr	2		-1.5302	0.6672	1098	-2.29	0.0220	
brthordr	1		0	
cohort		1	-3.2685	0.6072	460	-5.38	<.0001	
cohort		0	0	
brthordr*cohort	4	1	-0.08082	2.0917	1098	-0.04	0.9692	
brthordr*cohort	4	0	0	
brthordr*cohort	3	1	-0.9942	1.2808	1098	-0.78	0.4378	
brthordr*cohort	3	0	0	
brthordr*cohort	2	1	0.3419	0.9261	1098	0.37	0.7120	
brthordr*cohort	2	0	0	
brthordr*cohort	1	1	0	
brthordr*cohort	1	0	0	

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
brthordr	3	1098	11.53	<.0001
cohort	1	460	30.02	<.0001
brthordr*cohort	3	1098	0.34	0.7961

The conditional (residual) variance estimates seen in the “Covariance Parameter Estimates” table can be compared to the marginal (total) variance estimates obtained from the random-effects ANOVA results to see variance explained at each level of the model. Note that it is the between-family variance that has been most reduced by the inclusion of the Level 1 predictors.

The “Type 3 Tests of Fixed Effects” table gives omnibus (multiple degrees of freedom) tests of the main effects and interaction in the model. The “Solution for Fixed Effects” table shows the individual effects of the coding variables. Within the later table, the reference categories for CLASS variables are indicated with estimates of zero and dots for significance tests.

Adding Upper-Level Predictors to the Model

We now include the mother's age when she had her first child as a Level 2 predictor of intercepts, to undo confounding between maternal age and birth order:

Level 1:

$$\begin{aligned} \text{Math}_{ij} &= \beta_{0j} + \beta_{1j}\text{Second}_{ij} + \beta_{2j}\text{Third}_{ij} + \beta_{3j}\text{Fourth}_{ij} + \beta_{4j}\text{Old}_{ij} + \\ &\quad \beta_{5j}\text{Second}_{ij} \times \text{Old}_{ij} + \beta_{6j}\text{Third}_{ij} \times \text{Old}_{ij} + \beta_{7j}\text{Fourth}_{ij} \times \text{Old}_{ij} + r_{ij} \\ r_{ij} &\sim N(0, \sigma^2) \end{aligned}$$

Level 2:

$$\begin{aligned} \beta_{0j} &= \gamma_{00} + \gamma_{01}\text{BirthAge}_j + u_{0j} & u_{0j} &\sim N(0, \tau_{00}) \\ \beta_{pj} &= \gamma_{p0} \text{ for } p > 0 \end{aligned}$$

Reduced-Form:

$$\begin{aligned} \text{Math}_{ij} &= \gamma_{00} + \gamma_{01}\text{BirthAge}_j + \gamma_{10}\text{Second}_{ij} + \gamma_{20}\text{Third}_{ij} + \gamma_{30}\text{Fourth}_{ij} + \\ &\quad \gamma_{40}\text{Old}_{ij} + \gamma_{50}\text{Second}_{ij} \times \text{Old}_{ij} + \gamma_{60}\text{Third}_{ij} \times \text{Old}_{ij} + \gamma_{70}\text{Fourth}_{ij} \times \text{Old}_{ij} + \\ &\quad u_{0j} + r_{ij} \end{aligned}$$

The corresponding MIXED syntax follows. Note that `brthage` is first mean-centered to simplify interpretation of the model intercept.

```
proc standard data=birth out=birthcent m=0;
  var brthage;
run;

proc mixed data=birthcent method=reml covtest cl order=data noclprint;
  class brthordr cohort mom_id;
  model math = brthordr cohort cohort*brthordr brthage /
            solution ddfm=bw;
  random intercept / subject=mom_id;
run;
```

Abridged results follow:

Dimensions	
Covariance Parameters	2
Columns in X	16
Columns in Z Per Subject	1
Subjects	2207
Max Obs Per Subject	4

Covariance Parameter Estimates								
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	Alpha	Lower	Upper
Intercept	mom_id	43.3318	4.1352	10.48	<.0001	0.05	36.2454	52.7324
Residual		101.65	3.9923	25.46	<.0001	0.05	94.2625	109.96

Solution for Fixed Effects								
Effect	brthordr	cohort	Estimate	Standard Error	DF	t Value	Pr > t	
Intercept			99.1133	0.5175	2205	191.51	<.0001	
brthordr	4		0.1172	1.2627	1098	0.09	0.9261	
brthordr	3		0.1692	0.8397	1098	0.20	0.8403	
brthordr	2		0.01720	0.6746	1098	0.03	0.9797	
brthordr	1		0	
cohort		1	-0.2755	0.6604	460	-0.42	0.6768	
cohort		0	0	
brthordr*cohort	4	1	-2.4697	2.0825	1098	-1.19	0.2359	
brthordr*cohort	4	0	0	
brthordr*cohort	3	1	-2.7984	1.2775	1098	-2.19	0.0287	
brthordr*cohort	3	0	0	
brthordr*cohort	2	1	-0.5900	0.9186	1098	-0.64	0.5208	
brthordr*cohort	2	0	0	
brthordr*cohort	1	1	0	
brthordr*cohort	1	0	0	
brthage			0.7325	0.06889	2205	10.63	<.0001	

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
brthordr	3	1098	1.35	0.2557
cohort	1	460	7.30	0.0071
brthordr*cohort	3	1098	1.84	0.1385
brthage	1	2205	113.07	<.0001

Note the increment to “Columns in X” associated with the inclusion of **brthage**. The omnibus test for **brthordr** in the “Type 3 Tests of Fixed Effects” table is now non-significant. Similarly, in the “Solution for Fixed Effects” table, the effect of each dummy variable for birth order is non-significant.

One dummy variable for the birth order by cohort interaction is significant, suggesting a significant difference between first and third born children in the older cohort, but this should be interpreted cautiously (or not at all) given the lack of a significant omnibus cohort by birth order interaction.

Beta Interferon Analysis

For this demonstration we use the `beta.sas7bdat` data file. The data has been provided courtesy of Dr. Gideon Koren, who runs the Motherisk Program at the University of Toronto, Hospital for Sick Children. Accompanying SAS code is in `betaint.sas`. Variables are:

- MID** A unique numeric identifier for each mother.
- Group** A nominal variable indicating which group the mother belongs to, Multiple Sclerosis – Took Beta-Interferon (B-Ifn), Multiple Sclerosis – Discontinued Medication (Disc), or Healthy Control (HCntrl).
- GestAge** Gestational age of the infant at delivery, in weeks.
- MGain_lb** Mother’s weight gain during pregnancy, in pounds.
- CWeight_lb** Infant’s birth weight, in pounds.

Exploratory Data Analysis

We will begin by examining the characteristics of the data. The raw data is shown below:

```
proc print data=demo.beta;
run;
```

Obs	MID	group	Gest Age	MGain_lb	CWeight_lb
1	1	Disc	38	20	7.76367
2	1	Disc	38	20	7.54453
3	1	Disc	38	20	7.79497
4	7	B-Ifn	38	24	7.63845
5	8	B-Ifn	40	32	6.44885
6	9	B-Ifn	32	21	5.35317
7	10	B-Ifn	38	24	7.51323
8	10	B-Ifn	38	24	6.57407
9	11	B-Ifn	38	30	7.32540
10	12	B-Ifn	40	45	7.76367
11	13	B-Ifn	38	30	8.20194
12	14	B-Ifn	40	30	5.69753
13	15	B-Ifn	38	-15	8.20194
14	16	B-Ifn	35	43	6.63668
15	17	Disc	38	41	8.17063
16	18	Disc	38	19	7.51323
17	18	Disc	37	25	8.38977
18	18	Disc	38	9	9.14109
19	19	Disc	38	33	8.32716
20	20	Disc	30	23	5.22795
21	21	Disc	38	.	6.38624
22	22	Disc	38	30	7.51323
23	24	Disc	41	.	8.20194
24	25	Disc	38	76	9.20370
25	25	Disc	38	45	8.13933
26	25	Disc	39	55	8.64021
27	26	Disc	35	18	6.57407
28	26	Disc	35	24	6.57407

29	28	HCntrl	38	9	8.76543
30	29	HCntrl	39	40	8.70282
31	30	HCntrl	38	60	8.13933
32	31	HCntrl	39	50	8.32716
33	32	HCntrl	39	23	8.13933
34	33	HCntrl	35	75	8.51499
35	34	HCntrl	38	43	8.45238
36	35	HCntrl	39	24	8.20194
37	36	HCntrl	37	22	6.68677
38	37	HCntrl	37	40	6.10450
39	38	HCntrl	39	75	9.14109
40	39	HCntrl	39	25	8.89065
41	40	HCntrl	39	25	7.57584
42	41	HCntrl	38	31	7.76367
43	43	HCntrl	38	118	9.01587
44	43	HCntrl	38	86	9.39153
45	44	HCntrl	39	28	9.89242
46	45	HCntrl	39	35	8.45238

Note that there is some missing data on the maternal weight gain covariate.

The extent of nesting within the data is shown here:

```
proc freq data=demo.beta noprint;
  table MID /out=MIDfreq;
run;
proc freq data=MIDfreq;
  table count;
run;
```

The FREQ Procedure				
Frequency Count				
COUNT	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1	31	83.78	31	83.78
2	3	8.11	34	91.89
3	3	8.11	37	100.00


The mean birth weights for the three groups are computed and graphed here:

```
proc means data=demo.beta N mean median min max;
  class group;
  var gestage mgain_lb cweight_lb;
run;

proc gchart data=demo.beta;
  vbar group /discrete type=mean sumvar=cweight_lb axis=axis1
    descending;
  pattern value=solid color=red;
  axis1 label=("CWeight_lb") minor=none;
run; quit;
```

group	Obs	Variable	N	Mean	Median	Minimum
B-Ifn	11	GestAge	11	37.7272727	38.0000000	32.0000000
		MGain_lb	11	26.1818182	30.0000000	-15.0000000
		CWeight_lb	11	7.0322671	7.3253968	5.3531746
Disc	17	GestAge	17	37.3529412	38.0000000	30.0000000
		MGain_lb	15	30.5333333	24.0000000	9.0000000
		CWeight_lb	17	7.7121071	7.7949735	5.2279541
HCntrl	18	GestAge	18	38.2222222	38.5000000	35.0000000
		MGain_lb	18	44.9444444	37.5000000	9.0000000
		CWeight_lb	18	8.3421174	8.4523810	6.1044974

group	Obs	Variable	Maximum
B-Ifn	11	GestAge	40.0000000
		MGain_lb	45.0000000
		CWeight_lb	8.2019400
Disc	17	GestAge	41.0000000
		MGain_lb	76.0000000
		CWeight_lb	9.2037037
HCntrl	18	GestAge	39.0000000
		MGain_lb	118.0000000
		CWeight_lb	9.8924162



Random-Effects ANOVA Model

Fitting a random-effects ANOVA model will allow us to determine the intraclass correlation, and to partition the variance in birth weights into within- and between-mother components.

Level 1:

$$CWeight_lb_{ij} = \beta_{0j} + r_{ij} \quad r_{ij} \sim N(0, \sigma^2)$$

Level 2:

$$\beta_{0j} = \gamma_{00} + u_{0j} \quad u_{0j} \sim N(0, \tau_{00})$$

Reduced-Form:

$$CWeight_lb_{ij} = \gamma_{00} + u_{0j} + r_{ij}$$

The corresponding SAS code is

```
proc mixed data=demo.beta method=reml noclprint;
  class MID;
  model CWeight_lb = / solution alpha=.05 ddfm=kr;
  random intercept / subject=MID vcorr;
run;
```

Note that no inference tests were requested on the variance components given the very few mothers with multiple siblings that provide information on this partitioning. In addition, the DDFM=KR option in the MODEL statement invokes the Kenward-Rogers method for estimating standard errors and degrees of freedom for the fixed effects of the model. Standard errors are adjusted for small-sample bias in the variance parameter estimates and degrees of freedom are approximated via the Satterthwaite procedure. This method of testing fixed effects is optimal for small samples.

Output:

Model Information		
Data Set	DEMO.BETA	
Dependent Variable	CWeight_lb	
Covariance Structure	Variance Components	
Subject Effect	MID	
Estimation Method	REML	
Residual Variance Method	Profile	
Fixed Effects SE Method	Kenward-Roger	
Degrees of Freedom Method	Kenward-Roger	
Dimensions		
Covariance Parameters		2
Columns in X		1
Columns in Z Per Subject		1
Subjects		37
Max Obs Per Subject		3

Number of Observations						
Number of Observations Read			46			
Number of Observations Used			46			
Number of Observations Not Used			0			
Iteration History						
Iteration	Evaluations	-2 Res Log Like	Criterion			
0	1	138.41377055				
1	2	132.32910598	0.01819786			
2	1	131.78710783	0.00278792			
3	1	131.71216977	0.00007216			
4	1	131.71037070	0.00000005			
5	1	131.71036943	0.00000000			
Convergence criteria met.						
Estimated V Correlation Matrix for MID 1						
Row	Col1	Col2	Col3			
1	1.0000	0.7733	0.7733			
2	0.7733	1.0000	0.7733			
3	0.7733	0.7733	1.0000			
Covariance Parameter Estimates						
Cov Parm	Subject	Estimate				
Intercept	MID	0.9488				
Residual		0.2781				
Fit Statistics						
-2 Res Log Likelihood			131.7			
AIC (smaller is better)			135.7			
AICC (smaller is better)			136.0			
BIC (smaller is better)			138.9			
Solution for Fixed Effects						
Effect	Estimate	Standard Error	DF	t Value	Pr > t	Alpha
Intercept	7.7456	0.1801	35.4	43.01	<.0001	0.05
Solution for Fixed Effects						
Effect	Lower	Upper				
Intercept	7.3802	8.1110				

The ICC is shown by the VCORR output to be .7733. The between-mothers variance is .9488 and the within-mother variance is .2781. The pooled average for birth weight is 7.7456 lbs.

Adding Upper-Level Predictors to the Model

We now examine whether the mean differences in birth weight that we observed previously are statistically significant when accounting for dependence within mothers.

Level 1:

$$CWeight_lb_{ij} = \beta_{0j} + r_{ij} \quad r_{ij} \sim N(0, \sigma^2)$$

Level 2:

$$\beta_{0j} = \gamma_{00} + \gamma_{01}B\text{-Ifn} + \gamma_{02}Disc + u_{0j} \quad u_{0j} \sim N(0, \tau_{00})$$

Reduced-Form:

$$CWeight_lb_{ij} = \gamma_{00} + \gamma_{01}B\text{-Ifn} + \gamma_{02}Disc + u_{0j} + r_{ij}$$

The corresponding SAS code is:

```
proc mixed data=demo.beta method=reml noclprint;
  class MID Group;
  model CWeight_lb = Group / solution alpha=.05 ddfm=kr;
  random intercept / subject=MID vcorr;
  estimate "b-ifn v. disc" group 1 -1 0 / e cl;
run;
```

Syntax Notes:

- The ESTIMATE statement: Used to conduct null hypothesis tests on linear combinations of parameter estimates. Given the way the model has been specified, we will obtain tests of the B-ifn mean versus HCntrl and the Disc mean versus HCntrl, but we have no test of B-ifn versus Disc. To contrast the B-ifn group from the Disc group, we need to test null hypothesis: $H_0: \gamma_{01} - \gamma_{02} = 0$. This follows from the fact that

$$\mu_{B\text{-Ifn}} = \gamma_{00} + \gamma_{01}$$

$$\mu_{Disc} = \gamma_{00} + \gamma_{02}$$

It then follows that $\mu_{B\text{-Ifn}} - \mu_{Disc} = (\gamma_{00} + \gamma_{01}) - (\gamma_{00} + \gamma_{02}) = \gamma_{01} - \gamma_{02}$.

- You designate the label for the test in quotes, then indicate which effects you are combining using the name of the predictor, and you provide weights for the effect. Here, the weights are 1, -1, and 0 for the three levels of the group variable.
- The E option asks MIXED to report the weights used to compute the linear combination (useful for checking that you specified the linear combination correctly).
- The CL option requests a 95% confidence interval for the linear combination.

Abridged output is shown here:

Estimated V Correlation Matrix for MID 1			
Row	Col1	Col2	Col3
1	1.0000	0.7198	0.7198
2	0.7198	1.0000	0.7198
3	0.7198	0.7198	1.0000

Covariance Parameter Estimates		
Cov Parm	Subject	Estimate
Intercept	MID	0.7107
Residual		0.2766

Solution for Fixed Effects							
Effect	group	Estimate	Standard Error	DF	t Value	Pr > t	Alpha
Intercept		8.3001	0.2399	35.1	34.59	<.0001	0.05
group	B-Ifn	-1.2688	0.3935	34.8	-3.22	0.0027	0.05
group	Disc	-0.7570	0.3862	31.6	-1.96	0.0588	0.05
group	HCntrl	0

Solution for Fixed Effects			
Effect	group	Lower	Upper
Intercept		7.8130	8.7872
group	B-Ifn	-2.0678	-0.4697
group	Disc	-1.5439	0.03002
group	HCntrl	.	.

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
group	2	32.8	5.52	0.0086

Coefficients for b-ifn v. disc		
Effect	group	Row1
Intercept		
group	B-Ifn	1
group	Disc	-1
group	HCntrl	

Estimates						
Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha
b-ifn v. disc	-0.5118	0.4346	32.1	-1.18	0.2475	0.05
Estimates						
Label	Lower	Upper				
b-ifn v. disc	-1.3969	0.3732				

Note that the variance associated with mothers has been reduced from .9488 to .7107, a reduction of 25%

$$1 - \frac{\hat{\tau}_{00C}}{\hat{\tau}_{00U}} = 1 - \frac{.71}{.95} = .25$$

The omnibus test of the group effect is significant, indicating that the average birth weights of the three groups differ in the population. More detail on this effect can be gleaned from the coefficient estimates in the “Solution for Fixed Effects” table. These intercept estimate indicates that the average birth weight for an infant born to a healthy mother is 8.3001 lbs. In comparison, the average birth weight for an infant of a mother with MS who takes beta-interferon is 1.2688 lbs lower. The average birth weight for an infant of a mother with MS who stopped taking beta-interferon prior to conception is .7570 lbs lower than the weight of infants of healthy mothers, a marginal difference.

The way the group variable was coded did not provide a comparison of the b-ifn and disc groups, hence we used the ESTIMATE statement to compute this contrast. The results indicate that the mean for the b-ifn group is .5118 units lower than the mean for the disc group, but that this difference is not statistically significant.

We will now evaluate whether these effects are maintained after including maternal weight gain and gestational age as covariates.

Adding Lower-Level Predictors to the Model

The covariates will be centered at the median values observed for healthy controls and entered at Level 1 as follows:

Level 1:

$$CWeight_lb_{ij} = \beta_{0j} + \beta_{1j}GestAge_{ij} + \beta_{2j}MGain_lb + r_{ij} \quad r_{ij} \sim N(0, \sigma^2)$$

Level 2:

$$\beta_{0j} = \gamma_{00} + \gamma_{01}B\text{-Ifn} + \gamma_{02}Disc + u_{0j} \quad u_{0j} \sim N(0, \tau_{00})$$

$$\beta_{1j} = \gamma_{10}$$

$$\beta_{2j} = \gamma_{20}$$

Reduced-Form:

$$CWeight_lb_{ij} = \gamma_{00} + \gamma_{01}B\text{-Ifn} + \gamma_{02}Disc + \gamma_{10}GestAge_{ij} + \gamma_{20}MGain_lb + u_{0j} + r_{ij}$$

To fit the model we must first perform the centering. The following SAS code computes the medians for healthy controls and then centers the covariates at those values (centering was performed with respect to the median given a couple of unusually high levels of maternal weight gain among healthy controls). By centering, the intercepts can be interpreted as the expected birth weight for an infant born at a typical gestational age to a mother who experienced the typical amount of weight gain for a healthy pregnancy.

```
proc means data=demo.beta N mean median min max; where group="HCntrl";
  var GestAge MGain_lb;
run;
data beta2; set demo.beta;
  GestAge = GestAge - 38.5;
  MGain_lb = MGain_lb - 37.5;
run;
```

The MEANS Procedure					
Variable	N	Mean	Median	Minimum	Maximum
GestAge	18	38.2222222	38.5000000	35.0000000	39.0000000
MGain_lb	18	44.9444444	37.5000000	9.0000000	118.0000000

We now fit the model as follows:

```
proc mixed data=beta2 method=reml noclprint;
  class MID Group;
  model CWeight_lb = GestAge MGain_lb Group / solution alpha=.05
    ddfm=kr;
  random intercept / subject=MID v vcorr;
  estimate "b-ifn v. disc" group 1 -1 0 / e cl;
run;
```

Abridged output is shown here:

Number of Observations	
Number of Observations Read	46
Number of Observations Used	44
Number of Observations Not Used	2

Note the loss of two observations due to missing data on the maternal weight gain covariate.

Covariance Parameter Estimates		
Cov Parm	Subject	Estimate
Intercept	MID	0.3291
Residual		0.3323

Note that these variance estimates can only roughly be compared to the corresponding estimates from previous models given the difference in sample size (loss of observations due to missing data). The between-mother variance is made smaller by including the Level 1 covariates (due to between-mother differences in maternal weight gain and gestational age). The within-mother variance is similar to before (actually appearing to increase a bit, but bear in mind that these estimates are not very stable given the small sample, and that the sample is not quite the same as before).

Solution for Fixed Effects							
Effect	group	Estimate	Standard Error	DF	t Value	Pr > t	Alpha
Intercept		8.3364	0.1993	31.3	41.82	<.0001	0.05
GestAge		0.2734	0.06965	30.7	3.93	0.0005	0.05
MGain_lb		0.008817	0.006631	31.4	1.33	0.1932	0.05
group	B-Ifn	-0.9902	0.3400	30.6	-2.91	0.0066	0.05
group	Disc	-0.1695	0.3537	20.4	-0.48	0.6369	0.05
group	HCntrl	0

Solution for Fixed Effects				
Effect	group	Lower	Upper	
Intercept		7.9299	8.7428	
GestAge		0.1313	0.4155	
MGain_lb		-0.00470	0.02233	
group	B-Ifn	-1.6840	-0.2964	
group	Disc	-0.9064	0.5674	
group	HCntrl	.	.	

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
GestAge	1	30.7	15.41	0.0005
MGain_lb	1	31.4	1.77	0.1932
group	2	23.4	4.57	0.0211

The omnibus test of the group effect is still significant. Of interest, the difference between the Disc and HCntrl groups has largely vanished. The difference between the B-Ifn and HCntrl group is smaller but still statistically significant.

Estimates						
Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha
b-ifn v. disc	-0.8207	0.3738	20.9	-2.20	0.0396	0.05
Estimates						
Label	Lower	Upper				
b-ifn v. disc	-1.5983	-0.04309				

The difference between the B-ifn and Disc groups is now statistically significant.

Within SAS, all pair-wise comparisons between levels of a CLASS variable can also be obtained another way, as shown in the code below:

```
proc mixed data=beta2 method=reml noclprint;
  class MID Group;
  model CWeight_lb = GestAge MGain_lb Group / solution alpha=.05
        ddfm=kr;
  random intercept / subject=MID;
  lsmeans Group / at (GestAge MGain_lb)=(0 0) diff adjust=Tukey;
  ods output lsmeans=adjgroupmeans;
run;
```

Syntax Notes:

- The LSMEANS statement: Used to compute “least squares” means generates conditional mean estimates, adjusted for the covariates.
 - The AT option is used to hold the covariates at specific values (the default is to hold them at their means). Here, they are held at zero, the median values for the healthy control group after centering.
 - The DIFF option requests tests of mean differences between pairs of levels of the CLASS variable.
 - The ADJUST option requests adjusted significance testing so that the family-wise α -level will be maintained at .05. This is an important advantage of the LSMEANS statement relative to the ESTIMATE statement, particularly when the classification variable has more than a few levels. Here, the Tukey-Kramer adjustment has been requested (other options include Bonferroni, Scheffe, Dunnett, etc)
- The ODS statement: Calls the Output Delivery System and is used for a variety of purposes.
 - ODS OUTPUT tells SAS to convert a table of output into an active dataset. This statement is followed by a key word for a table of output (“lsmeans”) and then the name of the dataset to be created (“adjgroupmeans”). In the present case, the conditional means are being output in order to graph them. Keywords for tables can be found in the user guide documentation, online help, or by using ODS TRACE.

The new output produced by LSMEANS is shown here:

Least Squares Means							
Effect	group	Gest Age	MGain_lb	Estimate	Standard Error	DF	t Value
group	B-Ifn	0.00	0.00	7.3462	0.2703	28.9	27.18
group	Disc	0.00	0.00	8.1668	0.2948	16	27.70
group	HCntrl	0.00	0.00	8.3364	0.1993	31.3	41.82

Least Squares Means		
Effect	group	Pr > t
group	B-Ifn	<.0001
group	Disc	<.0001
group	HCntrl	<.0001

Differences of Least Squares Means								
Effect	group	_group	Gest Age	MGain_lb	Estimate	Standard Error	DF	t Value
group	B-Ifn	Disc	0.00	0.00	-0.8207	0.3738	20.9	-2.20
group	B-Ifn	HCntrl	0.00	0.00	-0.9902	0.3400	30.6	-2.91
group	Disc	HCntrl	0.00	0.00	-0.1695	0.3537	20.4	-0.48

Differences of Least Squares Means						
Effect	group	_group	Pr > t	Adjustment	Adj P	
group	B-Ifn	Disc	0.0396	Tukey-Kramer	0.0929	
group	B-Ifn	HCntrl	0.0066	Tukey-Kramer	0.0204	
group	Disc	HCntrl	0.6369	Tukey-Kramer	0.8817	

The conditional means (adjusted for the covariates) are 7.3462, 8.1668, 8.3364 for the B-Ifn, Disc and HCntrl groups, respectively. Note that after adjustment for multiple comparisons, the B-Ifn versus Disc difference is non-significant. The conditional means are now in the dataset **adjgroupmeans**. This data set is shown here:

```
proc print data=adjgroupmeans;
run;
```

Obs	Effect	group	Gest Age	MGain_lb	Estimate	StdErr	DF	tValue	Probt
1	group	B-Ifn	0.00	0.00	7.3462	0.2703	28.9	27.18	<.0001
2	group	Disc	0.00	0.00	8.1668	0.2948	16	27.70	<.0001
3	group	HCntrl	0.00	0.00	8.3364	0.1993	31.3	41.82	<.0001

We can now plot these means as shown here:

```
proc gchart data=adjgroupmeans;  
  vbar group /discrete type=mean sumvar=estimate axis=axis1 descending;  
  pattern value=solid color=red;  
  axis1 label=("CWeight_lb") minor=none value=("0" "1" "2" "3" "4" "5"  
                                               "6" "7" "8" "9");  
run; quit;
```

The resulting plot can be compared to the prior plot of unadjusted (raw) group means to see the impact of controlling for the covariates.

